# 8. A Zonal Energy Simulation Model for Cities and Environment

Howard T. Odum

The zones of a city may be represented as a cross section from the supporting rural areas on the left to the concentrated center on the right. Energy and EMERGY for each zone of the a city may be presented in this one dimension and related to the concepts and hypotheses (Figure 1). One hypothesis is that the activities and functions of the city are organized according to the transformity, ranging from low, rural values on the left to high values for concentrated information on the right (Chapter 2).

In order to explore the consequences of the theory of Zonal City organization, the complexities of two dimensional landscapes were reduced for computer simulation to a chain of hierarchical blocks, each representing an aggregated concentric zone as measured from the dispersed rural areas to the center of the city (Figure 1). Then a simulation model was prepared based on the hierarchical principle that each block maximizes its function by interacting with the zones on either side, pumping land back and forth in proportion to the assets developed in that zone. In this chapter, the properties of a unit model of one block representing one zone are given. Preliminary simulations were made of 5 zonal blocks together (not included in this chapter).

The sketch in Figure 1 summarizes the base model and the way the zones are connected. In each unit, the production function generates zonal assets and population in proportion to the product of local environmental input (times its area), times flow from assets from the outer zone, times feedback from the next higher zone. In later versions a population stock was added to each of the 5 zonal units.

Figure 2 contains the unit model for one zone in more detail including a population storage. Although the letters in the diagram are for the agricultural zone, the results would be similar for other sectors. Equations are given as derived from the energy systems model. Numbers for calibration are written on the pathways. These numbers were entered into the calibration work sheet in Table 1 so that coefficients for the equations could be calculated. These equations were included in a program in BASIC that simulates the model. The program is listed as Table 2.




Figure 1. Main features of a simulation model of aggregated zones of a city organized as a serial energy hierarchy converging from left to right. Each zone is represented by one autocatalytic block.




PR = Ra*B*Ga*Ag

Jag = k16*Ag    Ga = pa*Jag    Ae = A - Ag - Ah

Ra = E*Aa -k10*Ra*B*Ga*Ag    Ra = E*Aa/(1 + k10*B*Ga*Ag)

dAg/dt = k11*Ra*B*Ga*Ag -k12*Ag -k19*Ag*Ah -k18*Ag*H -k20*Ag*Ae -k16*Ag

dAa/dt = k5*Ae*Ag +k14*Ag*Ae -k4*B*Aa -k15*Aa*H

dAh/dt = k15*Aa*H - k14*Ah*Ag

Figure 2. Details and equations for one unit in the zonal simulation model of landscape. Abbreviations: E environmental inflow; Ae area of land in environmental zone (undeveloped areas and parks); Aa agriculture-aquaculture area; Ah Area of suburban housing; nPd, nP, and nPu are populations resident in these zones; B, Ag, and H are the structural storages, biomass, agricultural assets, housing; pa price of outside inputs; Ra local environmental resource unused.

Table 1

URBNUNIT.WK1

| | | | | | | | |
|---|---|---|---|---|---|---|---|
| | Sources: | Math | Calib: | Side | Units | Coefficients: | |
| | Environmental inflow | E = | 1 | E6 | m3/km2/yr | | |
| | Environ. Remainders: | Environ. area Ra = | 0.1 | E6 | m3/km2/yr | | |
| | Imports(fuels,elect.,G&S) | Relative price pa = | 20 | E-6 | | | |
| | Downscale Assets | B = | 1 | E7 | m2 assets | | |
| | Upscale Assets | H = | 1 | E7 | m2 assets | | |
| | Total Area available | A = | 60 | E6 | m2 land | | |
| | Down scale area | Ae = | 20 | E6 | m2 land | | |
| | Area in use | Ag = | 20 | E6 | m2 land | | |
| | Upscale area | Ah = | 20 | E6 | m2 land | | |
| | | | | | | | |
| | Storages | | | | | | |
| | Use area | Aa = | 20 | E4 6 | m2 assets | | |
| | Assets | Ag = | 1 | E4 | m2 assets | | |
| | G & Serv avail. to Develop. | Ga= | 1.00E+00 | | (normalized) | | |
| | | | | | | | |
| | Flows | | | | | | |
| | E use by assets | k10*Ra*B*Ga*Ag = | 9.00E-01 | E6 | m3/km2/yr | k10= | 0.45 |
| | Production | k11*Ra*B*Ga*Ag= | 3.00E-01 | E4 | m2 assets | k11= | 0.15 |
| | Assets depreciation | k12*Ag | 1.00E-01 | E4 | m2 assets | k12= | 0.005 |
| | G&S used | k13*Ga = | 1.00E+00 | ? | | k13= | 1 |
| | Land moved from upscale | k14*Ah*Ag= | 1.00E-01 | E6 | m2 land | k14= | 0.005 |
| | Land moved upscale | k15*Aa*H= | 1.00E-01 | E6 | m2 land | k15 | 0.005 |
| | Product export | Jag=k16*Ag= | 5.00E-02 | E4 | Asset equival. | k16= | 0.05 |
| | Ag Land Aa to Env. area Ae | k4*B*Aa = | 1.00E-01 | E6 | m2/yr | k4= | 0.005 |
| | Env. Land Ae to agr. Aa | k5*Ag*Aa = | 1.00E-01 | E6 | m2/yr | k5= | 0.005 |
| | Assets used by upscale systen | k18*Ag*H= | 5.00E-02 | E6 | m2 land | k18= | 0.0025 |
| | Assets use for upscale land | k19*Ag*As= | 5.00E-02 | E4 | m2 assets | k19= | 0.000125 |
| | Assets use for downscale land | k20*Ag*Ae= | 5.00E-02 | E4 | m2 assets | k20= | 0.000125 |

Table 2
Simulation Program in BASIC

```
10 REM Macintosh
20 'URBNUNIT.BAS  (model of one Zone  of landscape & City)
25 REM  H.T.Odum & Shu-li Huang 1994
30 SCREEN 1,0:COLOR 0,1
33 X =0:REM  X=1 prints bars; X=0 prints time graphs
45 IF X =1 GOTO 57
47 LINE (0,0)-(240,180),3,B
50 LINE (0,70)-(240,70),3
55 IF X=0 GOTO 62
57 LOCATE 13,1
59 PRINT "            B     Ag     H
60 LOCATE 1,1
62 REM Sources
64 H = 1
66 B = 1
68 pa = .1
70 B = 1
80 Ae = 20
90 Ah = 20
100 E = 1
80 REM Starting Conditions
82 A = 60
86 Aa = 20
94 Ag = 1
96 Ah = 20
102 Ga = 1
110 REM Scaling factors
114 Ag0 = 10
116 Aa0 = 1
120 Ah0 = 1
122 dt = .5
124 T0 = 1
126 B0 =.1
128 H0 = .1
130 REM Coefficients
140 k4 = .005
142 k5 = .005
150 k10 = .45
152 k11 = .15
154 k12 = .005
156 k13 = .005
158 k14 = .005
160 k15 = .005
```

```
162 k16 = .05
166 k18 = .0025
168 k19 = .000125
170 k20 = .000125
300 Ae = A -Aa- Ah
302 IF Ae<0 THEN Ae = 0
303 Ra = E*Aa /(1+ k10*B*Ga*Ag)
318 Ga = pa*Jag
321 Jag = k16*Ag
325 dAa = k5*Ae*Ag +k14*Ag*Ah - k4*B*Aa -k15*Aa*H
327 DAh =k15*Aa*H  -k14*Ah*Ag
340 dAg = k11*Ra*B*Ga*Ag - k12*Ag - k19*Ag*Ah -k18*Ag*H
-k20*Ag*Ae - Jag
400 Aa =Aa + dAa*dt
403 IF Aa <0 THEN Aa = 0
406 Ah = Ah + DAh * dt
409 IF Ah <0   THEN Ah = 0
510 Ag = Ag +dAg *dt
515 IF Ag <.00001 THEN Ag = .00001
600 T = T + dt
5000 REM MC
5010 REM Bar graph
5020 REM PLOTTING graphs:
5030 IF X = 0 GOTO 5200
5035 LINE (0,0)-(240,180),,B
5040 LINE (50,180)-(50,180-B/B0),3
5050 LINE (50,180-B/B0)-(80,180-B/B0),3
5060 LINE (80,180)-(80,180-B/B0),3
5065 LINE (80,180)-(80,180-Ag/Ag0),3
5070 LINE (80,180-Ag/Ag0)-(110,180-Ag/Ag0),3
5080 LINE (110,180)-(110,180-Ag/Ag0),3
5090 LINE (110,179-H/H0)-(140,179-H/H0),3
5100 LINE (140,180)-(140,180-H/H0),3
5150 LOCATE 1,25:PRINT "yrs"
5160 LOCATE 1,15: PRINT T
5180 IF X = 1 GOTO 300
5200 PSET (T/T0,179-Ag/Ag0),3
5300 PSET (T/T0,70 - Aa/Aa0),3
6000  IF T/T0 < 240 GOTO 300
7000  IF pa <30 THEN pa = pa +1
7050  IF pa>30 GOTO 7200
7060 T = 0: Ag = .1: Aa = 20: Ah = 20
7100 GOTO 300
7200  STOP
```

A simulation of the unit model was made holding the assets and populations of the zones on either side constant. Land area for the 3 sectors is held constant, being pulled into the sector with the stronger assets. In the simulation in Figure 3, the price of the inputs of fuels goods and services is varied. When the price is high, the assets within the zone do not develop, and the area is drawn away into the surrounding zones. When the price is lower, there is enough inputs to cause growth of the sectors assets up to the point where they are limited by the local environmental resources with which they must interact in the production function and by land available for expansion in the surrounding zones.

Not reported here are preliminary results of simulating the 5 unit model with outside inputs only to some zones, using normalized data starting with a rural landscape in a wilderness state. Gradually the other zones develop, each pulling area from the less central area to its left.




Figure 3. Simulation response of the unit model diagrammed in Figure 3 when pa (price of fuels-goods-services) is varied and the Assets and populations in adjacent zones are constant.